Лабораторная работа 3. Численное решение нелинейных уравнений

Задания:

1. Постановка задачи и интервалы изоляции корней
2. Метод простой итерации
3. Метод касательных
4. Метод хорд

ПРИМЕРНОЕ РАСПОЛОЖЕНИЕ РЕШЕНИЯ НА ЛИСТЕ ПОКАЗАНО В ПРИЛОЖЕНИИ.

Постановка задачи и интервалы изоляции корней

Найти наименьший положительный корень уравнения

$$\boxed{3\sin\left(\sqrt{x}\right)+0{,}35x-3{,}8=0}$$

с точностью ε=0,001.

Запишите условие задачи, как показано на рисунке.

	A	B	C	D	E	F	G	H	I	J	K	L
1	***Решение нелинейных уравнений***											
2	Задание: найти наименьший корень уравнения											
3	$\boxed{3\sin\left(\sqrt{x}\right)+0{,}35x-3{,}8=0}$											
4												
5	**Уравнение**		3*sin(x^0,5)+0,35*x-3,8			Производная			Вторая производная			
6			ε=	0,001		3*cos(x^0,5)/(2*x^0,5)+0,35			-3*sin(x^0,5)/(4*x)-3*cos(x^0,5)/(4*x^1,5)			

Интервалы изоляции корней определим табличным способом: составим в MS Excel таблицу, содержащую значения функции, ее первой и второй производной при x = 1, 2, …, 5 и построим точечную диаграмму f(x).

Отступите одну строчку от исходных данных, запишите подзаголовок: «Интервалы изоляции корней». Затем составьте таблицу, как показано на рисунке. Значения x подбираются таким образом, чтобы выявить все возможные корни уравнения (т.е. шаг должен быть достаточно мелким).

Значение x = 0,01 выбрано потому, что 0 не входит в область определения производных рассматриваемой функции.

Интервалы изоляции корней

x	f(x)	f'(x)	f''(x)
0,01	-3,497	15,2751	-753,741
1	-0,92559	1,16045	-1,03633
2	-0,1367	0,5154	-0,41176
3	0,21108	0,21095	-0,22358
4	0,32789	0,03789	-0,13148
5	0,31025	-0,06408	-0,0766



f(x) =3*SIN(A10^0,5)+0,35*A10-3,8

f'(x) =3*COS(A10^0,5)/(2*A10^0,5)+0,35

f''(x) =-3*SIN(A10^0,5)/(4*A10)-3*COS(A10^0,5)/(4*A10^1,5)

<u>Примечание</u>: ячейка A10 содержит первое значение x.

Проверим выполнение необходимого условия существования и достаточного условия единственности корня на отрезке с помощью функции ЕСЛИ:

а) корень существует, если $f(x_i)*f(x_{i+1})<0$;

б) корень единственный, если $f'(x_i)*f'(x_{i+1})>0$.

=ЕСЛИ(И(B10*B11<0;C10*C11>0);"да";"нет")

Результат: на отрезке [2, 3] имеется единственный корень. Выделите этот отрезок цветом.

Метод простой итерации (релаксации)

Найдем корень уравнения на отрезке [2, 3] методом релаксации.

Составим уравнение для итерационного процесса (i – номер итерации):

$x_{i+1} = x_i - c \cdot f(x_i)$;

c=2/(M+m),

где M=max(f'(x)), m=min(f'(x)) на рассматриваемом отрезке.

Так как на отрезке [2, 3] производная не меняет знак, можно считать, что максимальное и минимальное ее значения будут на концах отрезка. Следовательно, M = 0,515

m = 0,21 (см. таблицу значений функции и производных).

Вычислим с:

	G	H	I
9	Метод простой итерации		
10	$x_{i+1} = x_i - c*f(x_i)$		
11	c=	=2/(C12+C13)	

В примере используются ссылки на вычисленные значения производной в таблице определения интервалов изоляции.

Итак, с = 2,753.

Теперь составим таблицу для итераций:

i	x	f(x)	$\lvert x_{i+1}-x_i \rvert<\varepsilon$
0	2,5	0,07484	

Здесь i – номер итерации.

В ячейку x записано начальное приближение: $x_0 = (2+3)/2$ –середина рассматриваемого отрезка.

При вычислении f(x) используем x из **этой** таблицы (адрес H13):

=3*SIN(H13^0,5)+0,35*H13-3,8.

В последнем столбце проверяется условие окончания итерационного процесса. В данном примере будем заканчивать итерации, когда разница между соседними значениями x (x_i и x_{i+1}) станет меньше ε. Для нулевой итерации не проверяем, для последующих используем такое выражение:

=ЕСЛИ(ABS(H14-H13)<$D$6;"да";"нет").

Во второй строке таблицы (i = 1) в x запишем формулу для итераций, используя ссылку на вычисленное с, а также x и f(x) из прошлой итерации:

$x_1 = x_0 - c \cdot f(x_0)$

или=H13-$H$11*I13.

Вторую строчку нужно копировать, пока в столбце с условием не получится «да».

Метод простой итерации			
$x_{i+1} = x_i - c*f(x_i)$			
c=	2,75347		
i	x	f(x)	$\|x_{i+1}-x_i\|<\varepsilon$
0	2,5	0,07484	
1	2,29393	-0,00186	нет
2	=H14-H	0,00021	нет
3	2,29848	-2,4E-05	да

Ответ – x=2,298 (выделите цветом).

Метод касательных (Ньютона)

Итерационная формула:

$$x_{i+1} = x_i - f(x_i)/f'(x_i).$$

Выбор начального приближения (см. рис. в конце раздела):

Для метода Ньютона начальное приближение выбирается по условию $x_0 = a$, если $f(a)*f''(a)>0$, иначе $x_0 = b$.

Используйте для выбора начального приближения функцию ЕСЛИ и ссылки на таблицу с интервалами изоляции.

Составим таблицу для итераций:

	L	M	N	O	P
9	Метод Ньютона				
10	$x_{i+1} = x_i - f(x_i)/f'(x_i)$				
11	x_0=	2			
12	i	x	f(x)	f'(x)	$\|x_{i+1}-x_i\|$
13	0	2	-0,1367	0,5154	

В ячейку x вставлена ссылка на x_0. При вычислении f(x) и f'(x) пользуемся приведенными выше формулами, подставляя в них x из **этой** таблицы.

Во второй строке таблицы (i = 1) в x запишем формулу для итераций, используя ссылки на x, f(x) и f'(x) из прошлой итерации:

=M13-N13/O13

В столбце $|x_{i+1}-x_i|$ составим условие окончания итерационного процесса, такое же, как в методе простой итерации.

Вторую строчку нужно копировать, пока в столбце с условием не получится «да».

Метод Ньютона				
$x_{i+1} = x_i - f(x_i)/f'(x_i)$		=ЕСЛИ(B12*D12>0;A12;A13)		
x_0=	2			
i	x	f(x)	f'(x)	$\|x_{i+1}-x_i\|$
0	2	-0,1367	0,5154	
1	2,26523	-0,01365	0,41546	нет
2	2,29808	-0,00018	0,40425	нет
3	2,29854	-3,5E-08	0,4041	да

Метод хорд

Перед тем, как приступать к итерациям, нужно проверить условие и выбрать итерационную формулу и начальное приближение:



Запишите название метода, а затем с помощью функции ЕСЛИ определите x_0. В нашем примере получается x_0 = b, поэтому для расчетов берем формулу в розовом квадрате.



	G	H	I	J	K	L	M	N
18	Метод хорд							
19	xx_{i+1} = a-f(a)*(x_i-a)/(f(x_i)-f(a))			x_0=	3	←	=ЕСЛИ(B12*D12>0;A13;A12)	
20	a=	2	f(a)=	-0,1367				
21	i	x	f(x)	|x_{i+1}-x_i|				

Выпишем также значения аи f(a), которые нам пригодятся для вычислений (ссылки на таблицу с интервалами изоляции).

Далее строим таблицу для реализации метода. Условие окончания итерационного процесса такое же, как в предыдущих методах.



a=	2	f(a)=	-0,1367	
i	x	f(x)	|x_{i+1}-x_i|	
0	3	0,21108		
1	=$H$20-$	0,03672	нет	=$H$20-$J$20*(H22-$H$20)/(I22-$J$20)
2	2,30984	0,00455	нет	
3	2,29987	0,00054	нет	
4	2,29869	6,3E-05	нет	
5	2,29855	7,4E-06	да	

Индивидуальное задание

Найти один из корней нелинейного уравнения методом касательных (Ньютона) с точностью ε = 0,001.

Вариант 1	Вариант 2	Вариант 3
$y(x)=e^{-x}-\cos(x)$	$y(x)=2x^2-e^x-x$	$y(x)=1+\sin(x)-\ln(x+1)$
Вариант 4	Вариант 5	Вариант 6
$y(x)=0,25x^3+x-2$	$y(x)=0,1x^2-x\ln(x)$	$y(x)=3x-4\ln(x)-5$
Вариант 7	Вариант 8	Вариант 9
$y(x)=3x-14+e^x-e^{-x}$	$y(x)=e^x-e^{-x}-2$	$y(x)=e^{x-1}-x^3-x$
Вариант 10		
$y(x)=x^2+3x-1$		

ПРИЛОЖЕНИЕ:

Примерное расположение решения на листе MSExcel.

Решение нелинейных уравнений

Задание: найти наименьший корень уравнения

$$3\sin\left(\sqrt{x}\right)+0{,}35x-3{,}8=0$$

Уравнение	3*sin(x^0,5)+0,35*x-3,8	Производная	Вторая производная
	ε= 0,001	3*cos(x^0,5)/(2*x^0,5)+0,35	-3*sin(x^0,5)/(4*x)-3*cos(x^0,5)/(4*x^1,5)

Интервалы изоляции корней

x	f(x)	f'(x)	f''(x)	Условие
0,01	-3,497	15,27506	-753,741	нет
1	-0,92559	1,160453	-1,03633	нет
2	-0,1367	0,515403	-0,41176	да
3	0,21108	0,210954	-0,22358	нет
4	0,327892	0,03789	-0,13148	нет
5	0,310247	-0,06408	-0,0766	



Метод простой итерации

$x_{i+1} = x_i - c*f(x_i)$

c= 2,753466

i	x	f(x)	$\lvert x_{i+1}-x_i \rvert<\varepsilon$
0	2,5	0,07484	
1	2,293932	-0,00186	нет
2	2,299065	0,000214	нет
3	2,298477	-2,4E-05	да

Метод Ньютона

$x_{i+1} = x_i - f(x_i)/f'(x_i)$

x_0= 2

i	x	f(x)	f'(x)	$\lvert x_{i+1}-x_i \rvert$
0	2	-0,1367	0,515403	
1	2,265233	-0,01365	0,415459	нет
2	2,298079	-0,00018	0,404251	нет
3	2,298536	-3,5E-08	0,404096	да

Метод хорд

$x_{i+1} = a-f(a)*(x_i-a)/(f(x_i)-f(a))$ x_0= 3

a= 2 f(a)= -0,1367

i	x	f(x)	$\lvert x_{i+1}-x_i \rvert$
0	3	0,21108	
1	2,393068	0,036721	нет
2	2,309839	0,004546	нет
3	2,299867	0,000538	нет
4	2,298693	6,32E-05	нет
5	2,298555	7,43E-06	да